QUESTION: What does it mean for a sequence of functions to converge pointwise?

$f_n \to f$ pointwise on domain D means that $\lim_{n\to\infty} f_n(x) = f(x)$ for all x in D.

(f_n: sequence; f: limit function)

EXAMPLE: let $f_n(x) = x^n$, $f(x) = 0$, and $D = [0,1)$ interval

$f_1(x) = x$

$f_2(x) = x^2$

$f_3(x) = x^3$

$\vdots$



FOR A SERIES: $f(x) = \sum_{n=1}^{\infty} f_n(x)$ means $f(x) = \lim_{M\to\infty} \sum_{n=1}^{M} f_n(x)$ for all $x \in D$

EXAMPLE: $\sum_{n=0}^{\infty} x^n = \frac{1}{1-x}$ on $(0,1)$ ← pointwise convergence of a series of functions

for example: $x = \frac{1}{2}$: $\sum_{n=0}^{\infty} \left(\frac{1}{2}\right)^n = \frac{1}{1-\frac{1}{2}}$

$1 + \frac{1}{2} + \frac{1}{4} + \frac{1}{8} + \frac{1}{16} + \cdots = 2$



FOURIER SERIES:

The Fourier series of $f(x)$ on the interval $-L \le x \le L$ is:

$$f(x) \sim a_0 + \sum_{n=1}^{\infty} a_n \cos\left(\frac{n\pi}{L}x\right) + \sum_{n=1}^{\infty} b_n \sin\left(\frac{n\pi}{L}x\right)$$

($\sim$: "has the Fourier series")

where $a_0 = \frac{1}{2L}\int_{-L}^{L} f(x)\,dx$, $a_n = \frac{1}{L}\int_{-L}^{L} f(x)\cos\left(\frac{n\pi}{L}x\right)dx$, $b_n = \frac{1}{L}\int_{-L}^{L} f(x)\sin\left(\frac{n\pi}{L}x\right)dx$

CONVERGENCE THEOREM:

* [If $f(x)$ and $f'(x)$ are continuous, with the possible exception of finitely many jump discontinuities,] on $-L \le x \le L$, then the Fourier series of f converges:

- to the periodic extension of $f(x)$, where this extension is continuous,

• to $\frac{1}{2}[f(x+) + f(x-)]$ if f has a jump discontinuity at x.

* Terminology: Haberman says "piecewise smooth"; others say "piecewise C^1"

EXAMPLE:





Fourier Series

Math 330

Download a copy of this notebook that you can run in Mathematica at https://www.ml-wright.org/teaching/math330f19/other/fourier_series_coefficients.nb.

Throughout this file, Fourier series coefficients are denoted as follows:

$$a_0 + \sum_{n=1}^{\infty} a_n \cos\left(\frac{n\pi}{L}x\right) + \sum_{n=1}^{\infty} b_n \sin\left(\frac{n\pi}{L}x\right)$$

The formulas for the coefficients are:

$$a_0 = \frac{1}{2L}\int_{-L}^{L} f(x)\,dx, \quad a_n = \frac{1}{L}\int_{-L}^{L} f(x)\cos\left(\frac{n\pi}{L}x\right)dx \text{ and } b_n = \frac{1}{L}\int_{-L}^{L} f(x)\sin\left(\frac{n\pi}{L}x\right)dx \quad \text{for } n > 1$$

---

1. $f(x) = x$

In[1]:=
```
f[x_] := x
```

The cosine coefficients a_n are all zero since $f(x)$ is an odd function.

The sine coefficients b_n are:

In[2]:=
```
b[n_] =
 Simplify[1/L * Integrate[f[x] * Sin[n * Pi * x / L], {x, -L, L}], Assumptions → n ∈ Integers]
```

Out[2]= $-\frac{2(-1)^n L}{n\pi}$

Plot of partial sum with $L = 1$:

In[4]:=
```
Plot[{x, Sum[(b[n] /. L → 1) * Sin[n * Pi * x], {n, 1, 1000}]},
 {x, -3, 3}, PlotRange → {-1.5, 1.5}]
```

Out[4]=



2. $f(x) = |x|$

In[∘]:=
```
f[x_] := Abs[x]
```

The sine coefficients b_n are all zero since $f(x)$ is an even function.

The constant coefficients a_0 is:

In[∘]:=
```
a[0] = 1/(2 L) * Integrate[f[x], {x, -L, L}, Assumptions → {L ∈ Reals, L > 0}]
```

Out[∘]= $\frac{L}{2}$

The cosine coefficients a_n are:

In[∘]:=
```
a[n_] = Simplify[1/L * Integrate[f[x] * Cos[n * Pi * x / L], {x, -L, L}],
  Assumptions → {n ∈ Integers, L > 0}]
```

Out[∘]= $\frac{2\left(-1+(-1)^n\right) L}{n^2 \pi^2}$

Plot of partial sum with $L = 1$:

In[∘]:=
```
Plot[{f[x], (a[0] /. L → 1) + Sum[(a[n] /. L → 1) * Cos[n * Pi * x], {n, 1, 100}]},
 {x, -3, 3}, PlotRange → {-0.5, 2}]
```

Out[∘]=



3. $f(x) = 3x - 1$

In[∘]:=
```
f[x_] := 3 x - 1
```

The constant coefficients a_0 is:

In[∘]:=
```
a[0] = 1/(2 L) * Integrate[f[x], {x, -L, L}]
```

Out[∘]= -1

The cosine coefficients a_n are zero:

In[∘]:=
```
a[n_] = Simplify[1/L * Integrate[f[x] * Cos[n * Pi * x / L], {x, -L, L}],
  Assumptions → {n ∈ Integers, L > 0}]
```

Out[∘]= 0

The sine coefficients b_n are:

In[∘]:=
```
b[n_] = Simplify[1/L * Integrate[f[x] * Sin[n * Pi * x / L], {x, -L, L}],
  Assumptions → {n ∈ Integers, L > 0}]
```

Out[∘]= $-\frac{6(-1)^n L}{n\pi}$

Plot of partial sum with $L = 1$:

In[∘]:=
```
Plot[{f[x], a[0] + Sum[(b[n] /. L → 1) * Sin[n * Pi * x], {n, 1, 100}]},
 {x, -3, 3}, PlotRange → {-5, 3}]
```

Out[∘]=



4. $f(x) = x^2$

In[∘]:=
```
f[x_] := x^2
```

The constant coefficients a_0 is:

In[∘]:=
```
a[0] = 1/(2 L) * Integrate[f[x], {x, -L, L}]
```

Out[∘]= $\frac{L^2}{3}$

The cosine coefficients a_n are:

In[∘]:=
```
a[n_] = Simplify[1/L * Integrate[f[x] * Cos[n * Pi * x / L], {x, -L, L}],
  Assumptions → {n ∈ Integers, L > 0}]
```

Out[∘]= $\frac{4(-1)^n L^2}{n^2 \pi^2}$

The sine coefficients b_n are zero since $f(x)$ is an even function.

Plot of partial sum with $L = 1$:

In[∘]:=
```
Plot[ {f[x], (a[0] /. L → 1) + Sum[ (a[n]  /. L → 1) * Cos[n * Pi * x], {n, 1, 100}]},
  {x, -3, 3}, PlotRange → {-0.5, 1.5}]
```

Out[∘]=



5. $f(x)$ = 1 if $x \geq 0$, -1 if $x < 0$

In[∘]:=
```
f[x_] := Piecewise[{{1, x ≥ 0}, {-1, x < 0}}]
```

The cosine coefficients a_n are zero because $f(x)$ is an odd function.

The sine coefficients b_n are:

In[∘]:=
```
b[n_] = Simplify[
  1 / L * Integrate[f[x] * Sin[n * Pi * x / L], {x, -L, L}, Assumptions → {L ∈ Reals, L > 0}],
  Assumptions → {n ∈ Integers}]
```

Out[∘]= $-\frac{2\left(-1+(-1)^{n}\right)}{n\,\pi}$

Plot of partial sum with $L = 1$:

In[∘]:=
```
Plot[ {f[x], Sum[b[n] * Sin[n * Pi * x], {n, 1, 80}]}, {x, -3.5, 3.5}]
```

Out[∘]=



6. $f(x)$ = | sin(x) |

For this problem, we will assume that L = π. (Otherwise, the integrals are messy.)

In[∘]:=
```
f[x_] := Abs[Sin[x]]
```

The constant coefficient a_0 :

In[∘]:=
```
a[0] = 1 / (2 Pi) * Integrate[f[x], {x, -Pi, Pi}]
```

Out[∘]= $\frac{2}{\pi}$

The cosine coefficients a_n if $n > 1$:

In[ ]:=
```
a[n_] =
 Simplify[1/Pi * Integrate[f[x] * Cos[n * x], {x, -Pi, Pi}], Assumptions → {n ∈ Integers}]
```

Out[ ]=
$$-\frac{2\left(1+(-1)^{n}\right)}{\left(-1+n^{2}\right)\pi}$$

The sine coefficients b_n are zero because $f(x)$ is an even function.

Plot of partial sum:

In[ ]:=
```
Plot[ {f[x], a[0] + Sum[a[n] * Cos[n * x], {n, 2, 20}]},
 {x, -3 Pi, 3 Pi}, Ticks → {Range[-3 Pi, 3 Pi, Pi], Automatic}]
```

Out[ ]=